

Winter severity influences spotted seatrout mortality in a southeast US estuarine system

Timothy A. Ellis*, Jeffrey A. Buckel, Joseph E. Hightower

*Corresponding author: taellis@ncsu.edu

Marine Ecology Progress Series 564: 145–161 (2017)

Supplement 1: OpenBUGS code for multistate capture-recapture model.

Supplement 2: OpenBUGS code for daily survival model.

Supplement 3: Relocation histories of telemetered spotted seatrout.

Supplement 4: Swimming speeds of telemetered spotted seatrout and bottlenose dolphins.

Supplement 1

OpenBUGS code for the multistate capture-recapture model used to estimate demographic rates of telemetered spotted seatrout; modified from Kéry and Schaub (2012).

```
model {
```

Parameters

```
# Z: Instantaneous total mortality rate between t and t+1  
# F: Instantaneous fishing mortality rate between t and t+1  
# M: Instantaneous natural mortality rate between t and t+1  
# E: Instantaneous emigration rate between t and t+1  
# p: Probability of being detected in the receiver array
```

States

```
# 1 Alive  
# 2 Natural Death  
# 3 Emigrated  
# 4 Harvest
```

Observations

```
# 1 Detected alive  
# 2 Detected natural mortality  
# 3 Detected emigrating  
# 4 Not Detected
```

Priors, constraints, and calculated values

```
for (t in 1:(Periods-1)){  
  lnF[t] ~ dunif(-10,1) # uninformative prior  
  lnM[t] ~ dunif(-10,1) # uninformative prior  
  lnE[t] ~ dunif(-10,1) # uninformative prior  
  F[t] <- exp(lnF[t])  
  M[t] <- exp(lnM[t])  
  E[t] <- exp(lnE[t])  
  Z[t] <- F[t]+M[t]+E[t]
```

```

S[t] <- exp(-Z[t])
Z_mort[t] <- F[t]+M[t] # total instantaneous mortality rate for fish
S_mort[t] <- exp(-Z_mort[t]) # discrete survival rate for fish
A_mort[t] <- 1-(exp(-Z_mort[t])) # discrete mortality rate for fish
}

p[1] <- 1 # Model conditioned on first capture, estimate separate p for remaining periods
for (t in 2:(Periods)){
  p[t] ~ dunif(0, 1)
}

# Define state-transition and observation matrices
for (i in 1:nFish){
  # Define probabilities of State (t+1) given State (t). First index is state at time t, next is
  # state at t+1
  for (t in first[i):(last[i]-1)){
    ps[1,i,t,1] <- S[t]
    ps[1,i,t,2] <- M[t]*(1-S[t])/Z[t]
    ps[1,i,t,3] <- E[t]*(1-S[t])/Z[t]
    ps[1,i,t,4] <- F[t]*(1-S[t])/Z[t]
    ps[2,i,t,1] <- 0
    ps[2,i,t,2] <- 1
    ps[2,i,t,3] <- 0
    ps[2,i,t,4] <- 0
    ps[3,i,t,1] <- 0
    ps[3,i,t,2] <- 0
    ps[3,i,t,3] <- 1
    ps[3,i,t,4] <- 0
    ps[4,i,t,1] <- 0
    ps[4,i,t,2] <- 0
    ps[4,i,t,3] <- 0
    ps[4,i,t,4] <- 1
  } #t

  for (t in first[i):(last[i])){
    # Define probabilities of Observed (t) given State (t). First index is state, last index is
    # observed
    po[1,i,t,1] <- p[t] # State=alive, detected alive
    po[1,i,t,2] <- 0 # State=alive, natural death
    po[1,i,t,3] <- 0 # State=alive, emigrated
    po[1,i,t,4] <- 1-p[t] # State=alive, not detected
    po[2,i,t,1] <- 0 # State=natural mortality, detected alive
    po[2,i,t,2] <- 1 # State=natural mortality, detected natural mortality
    po[2,i,t,3] <- 0 # State=natural mortality, detected emigration
    po[2,i,t,4] <- 0 # State=natural mortality, not detected
    po[3,i,t,1] <- 0 # State=emigrated, detected alive
    po[3,i,t,2] <- 0 # State=emigrated, detected natural mortality
    po[3,i,t,3] <- 1 # State=emigrated, detected emigration
    po[3,i,t,4] <- 0 # State=emigrated, not detected
    po[4,i,t,1] <- 0 # State=harvested, detected alive
    po[4,i,t,2] <- 0 # State=harvested, detected natural mortality
    po[4,i,t,3] <- 0 # State=harvested, detected emigration
    po[4,i,t,4] <- 1 # State=harvested, not detected
  } #t
} #i

```

Likelihood

```
for (i in 1:nFish){
  for (t in 1:first[i]-1) {Alive[i,t]<-0}
  z[i,first[i]] <- 1 # Individuals have known status (alive) at first occasion in study
  Alive[i, first[i]] <- 1
  for (t in (first[i]+1):last[i]){
    z[i,t] ~ dcat(ps[z[i,t-1], i, t-1,]) # State process: draw State (t) given State (t-1)
    Alive[i,t] <- step(-z[i,t]+2) # Should be 1 for z=1, 0 for z=2 or 3
  } #t
  for (t in last[i]+1:Periods){Alive[i,t]<- 0}
  for (t in first[i]:last[i]){
    y[i,t] ~ dcat(po[z[i,t], i, t,]) # Observation process: draw Observed (t) given State (t)
  } #t
} #i
for (t in 1:Periods){FishAtRisk[t] <- sum(Alive[,t])}
}
```

Reference

Kéry M, Schaub M (2012) Bayesian population analysis using WinBUGS: a hierarchical perspective. Academic Press, Waltham, MA

Supplement 2

OpenBUGS code for the daily survival model that used logistic regression to predict the daily natural mortality rate of telemetered spotted seatrout as a function of mean daily water temperature; modified from McCarthy (2007). The known fates (alive, natural mortality, fishing mortality) were assumed to follow a multinomial distribution (see Friedl et al. 2013).

```
model {  
  
  # Parameters  
  # TelM: Instantaneous daily natural mortality rate  
  # TelF: Instantaneous daily fishing mortality rate  
  # WTemp: Mean daily water temperature (degrees Celsius)  
  # b[1]: intercept for logistic function  
  # b[2]: slope for logistic function  
  
  # Priors  
  b[1] ~ dnorm(0, 1.0E-6) # uninformative prior  
  b[2] ~ dnorm(0, 1.0E-6) # uninformative prior  
  for (i in 1:493){  
    lnTelF[i] ~ dunif(-10,1) # uninformative prior  
    TelF[i] <-exp(lnTelF[i])  
  }  
  
  # Model Structure and Likelihood  
  for (i in 1:493){  
    AtRisk[i] <- sum(X[i,1:3]) # total number of fish at risk  
    TelM[i] <- 1 / (1 + exp(-(b[1]+b[2]*WTemp[i])))  
    S[i] <- exp(-TelF[i]-TelM[i]) # daily survival  
    p[i, 1] <- S[i]  
    p[i,2] <- TelM[i]*(1-S[i])/(TelF[i]+TelM[i])  
    p[i,3] <- 1-sum(p[i,1:2]) # Probability of not being seen again (=harvested)  
    X[i,1:3] ~ dmulti(p[i,1:3], AtRisk[i]) # X matrix known fates are 1 = alive, 2 = natural  
      mortality, and 3 = fishing mortality  
  }  
  
  # Calculated values  
  for (i in 2:28){  
    M_temp[i] <- 1 / (1 + exp(-(b[1]+b[2]*i))) # predicted daily natural mortality  
  }  
}
```

References

McCarthy MA (2007) Bayesian methods for ecology. Cambridge University Press, Cambridge

Friedl SE, Buckel JA, Hightower JE, Scharf FS, Pollock KH (2013) Telemetry-based mortality estimates of juvenile spot in two North Carolina estuarine creeks. *Trans Am Fish Soc* 142:399-415

Supplement 3




Figure S1. Relocation histories for 37 telemetered spotted seatrout released in 2 adjacent tributaries of the Pungo River, North Carolina, USA, from December 2009 to March 2010. Each row characterizes the daily automated and manual detections and assumed fates for a tagged individual, including a 7-d censorship period following postsurgical release. Fates are coded as natural mortality determined from stationary transmitter (NM1), natural mortality determined from removal of transmitter during cold temperatures (NM2), unconfirmed fishing mortality (UF), immigration (I), and emigration (E).




Figure S2. Relocation histories for 10 telemetered spotted seatrout released in 2 adjacent tributaries of the Pungo River, North Carolina, USA, from November to December 2010. Each row characterizes the daily automated and manual detections and assumed fates for a tagged individual, including a 7-d censorship period following postsurgical release. Fates are coded as surgery-related mortality (SM), natural mortality determined from stationary transmitter (NM1), natural mortality determined from removal of transmitter during cold temperatures (NM2), immigration (I), and emigration (E).



Figure S3. Relocation histories for 26 telemetered spotted seatrout released in 2 adjacent tributaries of the Pungo River and for 45 telemetered spotted seatrout released in 2 adjacent tributaries of the Neuse River, North Carolina, USA, from November 2011 to May 2012. Each row characterizes the daily automated and manual detections and assumed fates for a tagged individual, including a 7-d censorship period following postsurgical release. Fates are coded as natural mortality due to possible predation event (NMP), unconfirmed fishing mortality (UF), confirmed fishing mortality (F), immigration (I), and emigration (E).

Supplement 4




Figure S4. Estimated speeds of telemetered spotted seatrout emigrating from study sites in the Pungo River and Neuse River estuaries, North Carolina, USA, compared to estimates reported by Bacheler et al. (2009) from the Neuse River estuary for bottlenose dolphins.

Reference

Bacheler NM, Buckel JA, Hightower JE, Paramore LM, Pollock KA (2009) A combined telemetry-tag return approach to estimate fishing and natural mortality rates of an estuarine fish. *Can J Fish Aquat Sci* 66: 1230–1244